Some statistics about Diophantine Equations

Andrej Bauer

November 3, 2006

We use *Mathematica* to compute statistics about solvability of small Diophantine equations in natural numbers.

Generation of terms

We are going to generate terms in the language of Peano arithmetic with zero, successor, addition and multiplication. First we define the successor function:

```
ClearAll[s]
s[n_] := n + 1
```

Functions **terms** generates terms of a given complexity with given variables (this will always include all terms of smaller complexity, up to equality). Our notion of complexity is the *depth* of the expression tree. At each step of generation we put the terms into canonical form and throw out duplicates.

```
ClearAll[combine, terms, termsUpTo]

combine[f_, lst_] :=
 Flatten[Function[t1, Function[t2, f[t1, t2]] /@
     Select[lst, OrderedQ[{t1, #1}] &]] /@ lst, 1]

terms[0, vars_] := terms[0, vars] = Prepend[vars, 0]
terms[n_, vars_] := terms[n, vars] =
  Union [
   Map[Expand@s[#] &, terms[n - 1, vars]],
   combine[Expand@Plus[#1, #2] &, terms[n - 1, vars]],
   combine[Expand@Times[#1, #2] &, terms[n - 1, vars]]
  ]
```

Generation of equations

Given a list of terms **lst**, we generate all equations $e_1 = e_2$, where we avoid taking both $e_1 = e_2$ and $e_2 = e_1$.

```
ClearAll[equations]
equations[n_, vars_] :=
 With[{lst = terms[n, vars]},
  Union @ Flatten[Function[e1, Function[e2, e1 == e2] /@
       Select[lst, OrderedQ[{e1, #}] &]] /@ lst, 1]]
```

We can use *Mathematica* function **FullSimplify** to rewrite equations in equivalent form. Many equations are reduced to the same form this way. In the following example the original equations are in the left column and the simplified ones in the right column.

```
{#, FullSimplify[#,
     Assumptions → {x ∈ Integers, y ∈ Integers, x ≥ 0, y ≥ 0}]} & /@
  Take[equations[2, {x, y}], {1, 4000, 200}] // TableForm
```

False	False
$2 == 3x$	False
$2x == x^4$	$2x == x^4$
$4x == 2x + 2x^2$	$x^2 == x$
$2x^2 == x y^2$	$2x^2 == x y^2$
$x^3 == 2 + 2y$	$x^3 == 2 + 2y$
$x^4 == x y + x^2 y$	$x^4 == x(1 + x) y$
$1 + 2x == 1 + 2x + x^2$	$x == 0$
$1 + 3x == x + x^2 + y$	$(-2 + x) x + y == 1$
$x + x^2 == x + y + x y + y^2$	$x^2 == y(1 + x + y)$
$1 + 2x + x^2 == x + x^2 + y$	$1 + x == y$
$y == 3y$	$y == 0$
$3y == x y + x^2 y$	$3y == x(1 + x) y$
$2x y == 2x + x y$	$x y == 2x$
$2x^2 y == x + 3y$	$2x^2 y == x + 3y$
$2y^2 == x^2 + x y$	$2y^2 == x(x + y)$
$2x y^2 == x + y^2$	$(-1 + 2x) y^2 == x$
$x y^3 == 1 + x + y$	$x y^3 == 1 + x + y$
$2 + y == 1 + x + y^2$	$x + y^2 == 1 + y$
$2x + y == 2y + 2x y$	$y + 2x y == 2x$

Function **reducedEquations** generates all reduced equations of given depth in given variables.

```
ClearAll[reducedEquations]
reducedEquations[n_, vars_] := Union@FullSimplify[equations[n, vars],
   Assumptions → {And @@ (# ∈ Integers && # ≥ 0 & /@ vars)}]
```

The number of equations of depth 2 in two variables is almost halved by this procedure:

```
equations[2, {x, y}] // Length
```

```
5049
```

```
reducedEquations[2, {x, y}] // Length
```

```
2792
```

Diophantine Equations

Mathematica can solve a Diophantine equation in natural numbers with the **FindInstance** function:

```
FindInstance[
 {(1 + x)^2 + (1 + y)^2 == (1 + z)^2 && x ≥ 0 && y ≥ 0 && z ≥ 0}, {x, y, z}, Integers]
```

```
{{x → 3, y → 2, z → 4}}
```

It returns an empty list when there is no solution:

```
FindInstance[{(2 x + 1)^2 == 4 y + 3 && x ≥ 0 && y ≥ 0}, {x, y}, Integers]
```

```
{}
```

It cannot deal with arbitrary equations, of course:

```
FindInstance[
 {(1 + x)^7 + (1 + y)^7 == (1 + z)^7 && x ≥ 0 && y ≥ 0 && z ≥ 0}, {x, y, z}, Integers]
```

```
— FindInstance::nsmet :
  The methods available to FindInstance are insufficient to find
    the requested instances or prove they do not exist. More…
```

```
FindInstance[
 {(1 + x)^7 + (1 + y)^7 == (1 + z)^7 && x ≥ 0 && y ≥ 0 && z ≥ 0}, {x, y, z}, Integers]
```

We define a function which returns **True**, **False** or **Maybe** depending on whether a given equation has a solution:

```
ClearAll[hasSolution]
Off[FindInstance::nsmet]
hasSolution[eq_, vars_] := Switch[
  FindInstance[eq && And @@ (# ≥ 0 & /@ vars), vars, Integers],
  {}, False,
  _List, True,
  _FindInstance, Print["Cannot decide: ", eq]; Maybe]
```

Examples of usage:

```
hasSolution[(1 + x)^2 + (1 + y)^2 == (1 + z)^2, {x, y, z}]
```

```
True
```

```
hasSolution[(2 x + 1)^2 == 4 y + 3, {x, y}]
```

```
False
```

```
hasSolution[(1 + x)^7 + (1 + y)^7 == (1 + z)^7, {x, y, z}]
```

```
Cannot decide: (1 + x)^7 + (1 + y)^7 == (1 + z)^7
```

```
Maybe
```

Now we can write a function which collects statistics about a set of equations:

```
ClearAll[stats]
stats[eqs_, vars_] := With[{r = hasSolution[#, vars] & /@ eqs},
  {{True, Count[r, True]},
   {False, Count[r, False]}, {Maybe, Count[r, Maybe]}}]
```

As an example, let us look at statistic for equations with term depth 1 in one variable:

```
terms[1, {x}]
```

```
{0, 1, x, 2 x, x^2, 1 + x}
```

```
equations[1, {x}] // ColumnForm
```

```
False
True
0 == x
0 == 2 x
0 == x^2
0 == 1 + x
1 == x
1 == 2 x
1 == x^2
1 == 1 + x
x == 2 x
x == x^2
x == 1 + x
2 x == x^2
2 x == 1 + x
x^2 == 1 + x
```

Of the above 16 equations, 11 have a solution and 5 do not:

```
stats[equations[1, {x}], {x}] // TableForm
```

```
True    11
False   5
Maybe   0
```

The same example with reduced equations:

```
reducedEquations[1, {x}] // ColumnForm
```

```
False
True
x == 0
x == 1
x == x^2
2 x == x^2
x^2 == 1 + x
```

Now we have only 7 equations left, two of which do not have a solution:

```
stats[reducedEquations[1, {x}], {x}] // TableForm
```

```
True    5
False   2
Maybe   0
```

Statistics

■ Depth 1

One variable:

```
stats[equations[1, {x}], {x}] // TableForm
```

```
True     11
False    5
Maybe    0
```

Two variables:

```
stats[equations[1, {x, y}], {x, y}] // TableForm
```

```
True     58
False    9
Maybe    0
```

Three variables:

```
stats[equations[1, {x, y, z}], {x, y, z}] // TableForm
```

```
True     178
False    13
Maybe    0
```

Four variables:

```
stats[equations[1, {x, y, z, w}], {x, y, z, w}] // TableForm
```

```
True     419
False    17
Maybe    0
```

Five variables:

```
stats[equations[1, {x, y, z, w, v}], {x, y, z, w, v}] // TableForm
```

```
True     841
False    21
Maybe    0
```

■ Depth 2

One variable:

```
stats[equations[2, {x}], {x}] // TableForm
```

True	175
False	124
Maybe	0

Two variables:

```
stats[equations[2, {x, y}], {x, y}] // TableForm
```

Cannot decide: $x^4 == 1 + x + y^2$

Cannot decide: $y^4 == 1 + x^2 + y$

Cannot decide: $1 + x^2 + y == y^2 + x y^2$

Cannot decide: $x^2 + x^2 y == 1 + x + y^2$

True	4308
False	737
Maybe	4

Three variables:

```
stats[equations[2, {x, y, z}], {x, y, z}] // TableForm
```

Cannot decide: $2 x^2 == 1 + 2 y + y^2$

Cannot decide: $2 x^2 == 1 + 2 z + z^2$

Cannot decide: $4 x^2 == 1 + x + y^2$

Cannot decide: $4 x^2 == 1 + x + z^2$

Cannot decide: $x^4 == 1 + x + y + x y$

Cannot decide: $x^4 == 1 + x + y^2$

Cannot decide: $x^4 == 1 + x + z + x z$

Cannot decide: $x^4 == 1 + x + z^2$

Cannot decide: $2 + 2 x == x y + y^2$

Cannot decide: $2 + 2 x == x y^2 + y^3$

Cannot decide: $2 + 2 x == x z + z^2$

Cannot decide: $2 + 2 x == x z^2 + z^3$

Cannot decide: $1 + x^2 == 2 y + 2 x y$

Cannot decide: $1 + x^2 == 2 y + y^2$

Cannot decide: $1+x^2==x+y+xy+y^2$

Cannot decide: $1+x^2==2y+2y^2$

Cannot decide: $1+x^2==2z+2xz$

Cannot decide: $1+x^2==2z+z^2$

Cannot decide: $1+x^2==x+z+xz+z^2$

Cannot decide: $1+x^2==2z+2z^2$

Cannot decide: $x+x^2==1+2y+y^2$

Cannot decide: $x+x^2==1+2z+z^2$

Cannot decide: $1+x+x^2==x+y+xy+y^2$

Cannot decide: $1+x+x^2==x+z+xz+z^2$

Cannot decide: $2x+x^2==1+y^2$

Cannot decide: $2x+x^2==1+z^2$

Cannot decide: $1+2x+x^2==2y^2$

Cannot decide: $1+2x+x^2==y+y^2$

Cannot decide: $1+2x+x^2==2z^2$

Cannot decide: $1+2x+x^2==z+z^2$

Cannot decide: $2x+2x^2==1+y^2$

Cannot decide: $2x+2x^2==1+z^2$

Cannot decide: $4xy==1+x+z^2$

Cannot decide: $4xy==1+y+z^2$

Cannot decide: $x^2y==1+y+z^2$

Cannot decide: $2y^2==1+2z+z^2$

Cannot decide: $4y^2==1+x^2+y$

Cannot decide: $4y^2==1+y+z^2$

Cannot decide: $xy^2==1+x+z^2$

Cannot decide: $y^4==1+x^2+y$

Cannot decide: $y^4==1+x+y+xy$

Cannot decide: $y^4 == 1 + y + z + y z$

Cannot decide: $y^4 == 1 + y + z^2$

Cannot decide: $1 + 2 x + y == x y + y^2$

Cannot decide: $1 + 2 x + y == y^2 + x y^2$

Cannot decide: $1 + 2 x + y == x y^2 + y^3$

Cannot decide: $1 + 2 x + y == x z^2 + y z^2$

Cannot decide: $1 + x^2 + y == x y + y^2$

Cannot decide: $1 + x^2 + y == y^2 + x y^2$

Cannot decide: $1 + x^2 + y == 4 y z$

Cannot decide: $1 + x^2 + y == y z^2$

Cannot decide: $2 + 2 y == y z + z^2$

Cannot decide: $2 + 2 y == y z^2 + z^3$

Cannot decide: $1 + x + 2 y == x z^2 + y z^2$

Cannot decide: $x^2 + x y == 1 + x + y^2$

Cannot decide: $1 + x + y + x y == x y z + x z^2$

Cannot decide: $1 + x + y + x y == x y z + y z^2$

Cannot decide: $x + x^2 + y + x y == 1 + y^2$

Cannot decide: $x + x^2 + y + x y == 1 + y + y^2$

Cannot decide: $2 x + 2 x y == 1 + y^2$

Cannot decide: $x^2 + x^2 y == 1 + x + y^2$

Cannot decide: $1 + y^2 == 2 z + 2 y z$

Cannot decide: $1 + y^2 == 2 z + z^2$

Cannot decide: $1 + y^2 == y + z + y z + z^2$

Cannot decide: $1 + y^2 == 2 z + 2 z^2$

Cannot decide: $1 + x + y^2 == 4 x z$

Cannot decide: $1 + x + y^2 == x z^2$

Cannot decide: $y + y^2 == 1 + 2 z + z^2$

Cannot decide: $1 + y + y^2 == y + z + y z + z^2$

Cannot decide: $2 y + y^2 == 1 + z^2$

Cannot decide: $1 + 2 y + y^2 == 2 z^2$

Cannot decide: $1 + 2 y + y^2 == z + z^2$

Cannot decide: $2 y + 2 y^2 == 1 + z^2$

Cannot decide: $4 x z == 1 + y^2 + z$

Cannot decide: $x^2 z == 1 + y^2 + z$

Cannot decide: $4 y z == 1 + x^2 + z$

Cannot decide: $y^2 z == 1 + x^2 + z$

Cannot decide: $4 z^2 == 1 + x^2 + z$

Cannot decide: $4 z^2 == 1 + y^2 + z$

Cannot decide: $z^4 == 1 + x^2 + z$

Cannot decide: $z^4 == 1 + y^2 + z$

Cannot decide: $z^4 == 1 + x + z + x z$

Cannot decide: $z^4 == 1 + y + z + y z$

Cannot decide: $1 + 2 x + z == x y^2 + y^2 z$

Cannot decide: $1 + 2 x + z == x z + z^2$

Cannot decide: $1 + 2 x + z == z^2 + x z^2$

Cannot decide: $1 + 2 x + z == x z^2 + z^3$

Cannot decide: $1 + x^2 + z == x z + z^2$

Cannot decide: $1 + x^2 + z == z^2 + x z^2$

Cannot decide: $1 + 2 y + z == x^2 y + x^2 z$

Cannot decide: $1 + 2 y + z == y z + z^2$

Cannot decide: $1 + 2 y + z == z^2 + y z^2$

Cannot decide: $1 + 2 y + z == y z^2 + z^3$

Cannot decide: $1 + y^2 + z == y z + z^2$

Cannot decide: $1 + y^2 + z == z^2 + y z^2$

Cannot decide: $1+x+2z==xy^2+y^2z$

Cannot decide: $1+y+2z==x^2y+x^2z$

Cannot decide: $x^2+xz==1+x+z^2$

Cannot decide: $1+x+z+xz==xy^2+xyz$

Cannot decide: $1+x+z+xz==xyz+y^2z$

Cannot decide: $x+x^2+z+xz==1+z^2$

Cannot decide: $x+x^2+z+xz==1+z+z^2$

Cannot decide: $2x+2xz==1+z^2$

Cannot decide: $x^2+x^2z==1+x+z^2$

Cannot decide: $y^2+yz==1+y+z^2$

Cannot decide: $1+y+z+yz==x^2y+xyz$

Cannot decide: $1+y+z+yz==x^2z+xyz$

Cannot decide: $y+y^2+z+yz==1+z^2$

Cannot decide: $y+y^2+z+yz==1+z+z^2$

Cannot decide: $2y+2yz==1+z^2$

Cannot decide: $y^2+y^2z==1+y+z^2$

True	38680
False	1963
Maybe	111

■ Depth 3

Depth 3 is quite large. We can only handle the one variable case:

```
stats[equations[3, {x}], {x}] // TableForm
```

True	26581
False	28026
Maybe	0